# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: pzl <alsoelp _at_ gmail>

pkgname=perl-devel-checklib
pkgver=1.16
pkgrel=1
pkgdesc="Module to check if other perl modules are installed"
arch=('any')
url='https://search.cpan.org/dist/Devel-CheckLib/'
license=('PerlArtistic' 'GPL')
depends=('perl')
makedepends=('perl-capture-tiny')
options=('!emptydirs')
source=("https://search.cpan.org/CPAN/authors/id/M/MA/MATTN/Devel-CheckLib-${pkgver}.tar.gz")
sha512sums=('919da046cdd1f48e0ccd8a7734a98f3144c0005c8303b02b43aeedcd2cfc148c78f502f26a80ea2612279e2becf4948b408d8cbd4b7cc514519c5954ae4139cb')

build() {
  cd Devel-CheckLib-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd Devel-CheckLib-$pkgver
  make test
}

package() {
  cd Devel-CheckLib-$pkgver
  make install DESTDIR="$pkgdir"
}
